# $Id: PKGBUILD 158895 2012-05-12 18:37:10Z ibiru $
# Maintainer: Ionut Biru <ibiru@archlinux.org>
pkgname=libvpx
pkgver=1.1.0
pkgrel=1
pkgdesc="The VP8 Codec SDK"
arch=('i686' 'x86_64')
url="http://www.webmproject.org/"
license=('BSD')
depends=('glibc')
makedepends=('yasm')
source=(http://webm.googlecode.com/files/$pkgname-v$pkgver.tar.bz2)
sha1sums=('356af5f770c50cd021c60863203d8f30164f6021')

build() {
    cd "$pkgname-v$pkgver"
    ./configure --prefix=/usr --enable-vp8 \
        --enable-runtime-cpu-detect \
        --enable-shared \
        --enable-postproc \
        --enable-pic \
        --disable-install-docs \
        --disable-install-srcs
    make
}

package() {
    cd "$pkgname-v$pkgver"
    make DIST_DIR="$pkgdir/usr" install
    install -D -m 0644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
